1 Initial values

1.1 Constants

```
> sigma:=.01;
```

$$\sigma := 0.01$$

1.2 Initial yield curve

```
> f:=M-> 0.1 + h1 + h2*M;
```

$$f := M \rightarrow 0.1 + \mathit{h1} + \mathit{h2}\, M$$

1.3 Process parameters

```
> X0:={t=0,c=1,c2=0,h1=0,h2=0};
```

$$\mathit{X0} := \{t = 0,\ c = 1,\ \mathit{h1} = 0,\ \mathit{h2} = 0,\ \mathit{c2} = 0\}$$

```
> da:=V->evalf(subs(X0,V));
```

$$\mathit{da} := V \rightarrow \mathrm{evalf}(\mathrm{subs}(\mathit{X0},\ V))$$

```
> da([t,c,r]);
```

$$[0.,\ 1.,\ r]$$

2 The Process

2.1 Required operators

```
> T := proc (Index,power,V)
> subs(Index=Index+power,V);
> simplify(%);
> end proc:
```

```
> B := proc (Index, power, V)
> 1/2 * subs(Index=Index + power, V) +
> 1/2 * subs(Index=Index - power, V);
> simplify(%);
> end proc:
```

2.2 Definining Theta

```
> Theta:= V->
> B(h1, sigma,
> T(h1, mu1,
> T(h2, mu2,
> T(c, f(t)*c,
> T(c2, f(t)*c2,
> T(t,1,
> V
> ))))));
```

$$\Theta := V \rightarrow \mathrm{B}(\mathit{h1},\ \sigma,\ \mathrm{T}(\mathit{h1},\ \mu 1,\ \mathrm{T}(\mathit{h2},\ \mu 2,\ \mathrm{T}(c,\ \mathrm{f}(t)\, c,\ \mathrm{T}(\mathit{c2},\ \mathrm{f}(t)\, \mathit{c2},\ \mathrm{T}(t,\ 1,\ V))))))$$

2.3 Finding the risk neutral drift

```
> unassign('mu1');
> unassign('mu2');
> Theta(c):
> L1:=da(%);
```

$$L1 := 1.100000000 + \mu 1$$

```
> Theta(Theta(c)):
> L2:=da(%);
```

$$L2 := 1.210100000 + 2.200000000\,\mu 2 + 2.\,\mu 2\,\mu 1 + 3.300000000\,\mu 1 + 2.\,\mu 1^2$$

```
> L:=solve(
> {L1=(1+f(t))*c,L2=(1+f(t))*(1+f(t+1))*c},
> {mu1,mu2}):
> L:=da(L);
```

$$L := \{\mu 1 = 0.,\ \mu 2 = -0.00004545454546\}$$

```
> assign(L);
```

3 Tests

3.1 Cash account

```
> F:=c;
> da(F);
> for i from 0 to 5 do
> F:=Theta(F):
> print('Theta'^(i+1)*c=da(%));
> end:
```

$$F := c$$

$$1.$$

$$\Theta\, c = 1.100000000$$

$$\Theta^2\, c = 1.210000000$$

$$\Theta^3\, c = 1.330999991$$

$$\Theta^4\, c = 1.464099909$$

$$\Theta^5\, c = 1.610509570$$

$$\Theta^6\, c = 1.771559560$$

3.2 CMO

```
> F:=c;
> da(F);
> for i from 0 to 7 do
> F:= Theta(T(c,-0.2,F)):
> print(da(F));
> end:
```

$$F := c$$

$$1.$$
$$0.9000000000$$
$$0.7900181818$$
$$0.6690945315$$
$$0.5361949592$$
$$0.390206008$$
$$0.229929196$$
$$0.054074993$$
$$-0.138743633$$

3.3 Option

```
> F:= piecewise(f(t)>0.15,1,0);
```

$$F := \begin{cases} 1 & 0. < -0.05 + \mathit{h1} + \mathit{h2}\, t \\ 0 & \mathit{otherwise} \end{cases}$$

```
> for i from 0 to 10 do
> F:=Theta(F):
> print('Theta'^(i+1) * Opt = da(F));
> end:
```

$$\Theta\, Opt = 0.$$
$$\Theta^2\, Opt = 0.$$
$$\Theta^3\, Opt = 0.$$
$$\Theta^4\, Opt = 0.$$
$$\Theta^5\, Opt = 0.$$
$$\Theta^6\, Opt = 0.01562500000$$
$$\Theta^7\, Opt = 0.007812500000$$
$$\Theta^8\, Opt = 0.03515625000$$
$$\Theta^9\, Opt = 0.01953125000$$
$$\Theta^{10}\, Opt = 0.05468750000$$
$$\Theta^{11}\, Opt = 0.03271484375$$

3.4 Variance

```
> F:=c;
> F1:=F;
> F2:=F^2;
```

$$F := c$$
$$\mathit{F1} := c$$
$$\mathit{F2} := c^2$$

```
> sqrt(Theta(F2)-Theta(F1)^2):
> da(%);
```

$$0.0100000$$

```
> sqrt(Theta(Theta(F2))-Theta(Theta(F1))^2):
> da(%);
```

$$0.02459611758$$

3.5 Optimization

```
> P0:= V -> T(c2,1,V);
```

$$P0 := V \to \mathrm{T}(c2,\, 1,\, V)$$

```
> P1:= V -> P0(Theta(V));
```

$$P1 := V \to \mathrm{P0}(\Theta(V))$$

```
> P2:= V -> P1(Theta(V));
```

$$P2 := V \to \mathrm{P1}(\Theta(V))$$

```
> P3:= V -> P2(Theta(V));
```

$$P3 := V \to \mathrm{P2}(\Theta(V))$$

```
> P:= V ->
> T(c2,-a1/da(P1(V)),
> T(c2,-a2/da(P2(V)),
> T(c2,-a3/da(P3(V)),
> Theta(
> T(c2, a1,
> Theta(
> T(c2, a2,
> Theta(
> T(c2, a3,
> V
> )))))))));
```

$$P := V \to \mathrm{T}(c2,\, -\frac{a1}{\mathrm{da}(\mathrm{P1}(V))}, \mathrm{T}(c2,\, -\frac{a2}{\mathrm{da}(\mathrm{P2}(V))},$$
$$\mathrm{T}(c2,\, -\frac{a3}{\mathrm{da}(\mathrm{P3}(V))},\, \Theta(\mathrm{T}(c2,\, a1,\, \Theta(\mathrm{T}(c2,\, a2,\, \Theta(\mathrm{T}(c2,\, a3,\, V)))))))))$$

```
> P(c2):
> da(%);
```

$$-0.0002727210000\, a2 - 0.0001999670000\, a1$$

```
> Risk:=da(P(c2^2)-P(c2)^2);
```

$$Risk := 0.01214499057\, a1^2 + 0.05510645560\, a3\, a1 + 0.04219481792\, a2\, a1$$
$$+ 0.06292839268\, a3^2 + 0.03674961003\, a2^2 + 0.09609455474\, a3\, a2$$
$$- 1.\, (-0.0002727210000\, a2 - 0.0001999670000\, a1)^2$$

```
> plot3d(subs(a2=0,Risk), a1=-1..1, a3=-1..1,axes=BOXED);
```




```
> solve({
> diff(Risk,a1)=0,
> diff(Risk,a3)=0},
> {a1,a3});
```

$$\{a1 = -0.7408758737\, a2,\ a3 = -0.4391301679\, a2\}$$